

EXCEL 常用函数公式及举例

一、相关概念

（一）函数语法

由函数名+括号+参数组成

例： 求和函数：SUM (A1,B2,...) 。参数与参数之间用逗号“,” 隔开

（二）运算符

1. 公式运算符：加 (+)、减 (-)、乘 (*)、除 (/)、百分号 (%)、乘幂 (^)
2. 比较运算符：大于 (>)、小于 (<)、等于 (=)、小于等于 (<=)、大于等于 (>=) 、不等于 (<>)
3. 引用运算符：区域运算符 (:)、联合运算符 (,)

（三）单元格的相对引用与绝对引用

例：

A1
\$A1 锁定第 A 列
A\$1 锁定第 1 行
\$A\$1 锁定第 A 列与第 1 行

二、常用函数

（一）数学函数

1. 求和 =SUM (数值 1, 数值 2, ……)
2. 条件求和 =SUMIF (查找的范围, 条件(即对象), 要求和的范围)

例：(1) =SUMIF(A1:A4, ">=200", B1:B4)

函数意思：对第 A1 栏至 A4 栏中，大于等于 200 的数值对应的第 B1 列至 B4 列中数值求和

(2) =SUMIF(A1:A4, "<300", C1:C4)

函数意思：对第 A1 栏至 A4 栏中，小于 300 的数值对应的第 C1 栏至 C4 栏中数值求和

3. 求个数 =COUNT（数值 1,数值 2,……）

例：（1） =COUNT(A1:A4) 函数意思：第 A1 栏至 A4 栏求个数

 （2） =COUNT(A1:C4) 函数意思：第 A1 栏至 C4 栏求个数

4. 条件求个数 =COUNTIF（范围,条件）

例：（1） =COUNTIF(A1:A4,"<>200"）

 函数意思：第 A1 栏至 A4 栏中不等于 200 的栏求个数

 （2） =COUNTIF(A1:C4,">=1000"）

 函数意思：第 A1 栏至 C4 栏中大于等 1000 的栏求个数

5. 求算术平均数 =AVERAGE（数值 1,数值 2,……）

例：（1） =AVERAGE(A1,B2)

 （2） =AVERAGE(A1:A4)

6. 四舍五入函数 =ROUND(数值,保留的小数位数)

7. 排位函数 =RANK（数值,范围,序别） 1-升序 0-降序

例：（1） =RANK(A1,A1:A4,1)

 函数意思：第 A1 栏在 A1 栏至 A4 栏中按升序排序，返回排名值。

 （2） =RANK(A1,A1:A4,0)

 函数意思：第 A1 栏在 A1 栏至 A4 栏中按降序排序，返回排名值。

8. 乘积函数 =PRODUCT（数值 1,数值 2,……）

9. 取绝对值 =ABS(数字)

10. 取整 =INT(数字)

（二）逻辑函数

条件函数：=IF（条件,执行条件,执行条件假）—可执行七层嵌套

例： A1=5 A2=3 A3=6 A4=7

=IF(A1>A2,1,0)=1（真）

=IF(A1<A2,1,0)=0（假）

=IF(A1>A2,IF(A3>A4,8,9),1)=9

（三）文本函数

1. 截取函数：=LEFT(文本,截取长度)—从左边截取

=RIGHT(文本,截取长度)—从右边截取

=MID(文本,开始位,截取长度)—从中间截取

2. 计算字符长度：LEN(文本)—文本的长度

3. 合并字符函数：CONCATENATE(文本1,文本2,...)

4. 在字符串中查找特定字符：FIND(文本,范围,数值)—数值表示查找第几个

=FIND("a","abcadeafga",2)=4

=FIND("a","abcadeafga",3)=7

=FIND("a","abcadeafga",4)=10

5. 比较两个字符是否完全相符：EXACT(文本1,文本2)

=EXACT("a","a")=TRUE

=EXACT("a","ab")=FALSE

=EXACT("a","A")=FALSE

6. 将数值转化为文本：TEXT(数值,参数)—参数一般为0

7. 将数值型字符转换为数值：VALUE(数值文本)

（四）判断公式

1. 把公式产生的错误显示为空

公式：C2=IFERROR(A2/B2, "")

说明：如果是错误值则显示为空，否则正常显示。

H4 fx =IFERROR(D4/G4*100, "")									
	A	B	C	D	E	F	G	H	I
1	2016年“开门红”存款完成情况通报表								
2	数据截止日期：2016年3月24日								
3	序号	机构号	机构名称	净增额	年初余额	本期余额	任务数	完成率	排名
4	1	5839	营业部	2,367	31,643	34,010	5,200	45.52	12

2. IF 多条件判断返回值

公式：C2=IF(AND(A2<500,B2="未到期"), "补款", "")

说明：两个条件同时成立用 AND，任一个成立用 OR 函数。

C2 fx =IF(AND(A2<500,			
	A	B	C
1	金额	是否到期	提醒
2	800	未到期	
3	200	已到期	
4	300	未到期	补款
5			

（五）统计公式

1. 统计两个表格重复的内容

公式：B2=COUNTIF(Sheet15!A:A,A2)

说明：如果返回值大于 0 说明在另一个表中存在，0 则不存在。

	A	B	C
1	姓名	查找重复	
2	A	0	
3	B	1	
4	C	1	
5	D	1	
6	G	0	
7	Y	0	

	A	B	C
1	姓名		
2	D		
3	C		
4	R		
5	Q		
6	W		
7	B		

2. 统计不重复的总人数

公式：C2=SUMPRODUCT(1/COUNTIF(A2:A8,A2:A8))

说明：用 COUNTIF 统计出每人的出现次数，用 1 除的方式把出现次数变成分母，然后相加。

	A	B	C	D	E	F	G
1	姓名		总人数				
2	A		4				
3	B						
4	C						
5	D						
6	A						
7	B						
8	D						

（六）求和公式

1. 隔列求和

公式：H3=SUMIF(\$A\$2:\$G\$2,H\$2,A3:G3)

或者：=SUMPRODUCT((MOD(COLUMN(B3:G3),2)=0)*B3:G3)

说明：如果标题行没有规则用第 2 个公式

	A	B	C	D	E	F	G	H	I
1	产品	1月	2月	3月	合计				
2		实际	计划	实际	计划	实际	计划	实际	计划
3	A	1	6	9	3	7	4	17	
4	B	9	7	6	5	2	3	7	
5	C	6	10	10	7	6	5	2	
6									
7									
8									
9									
10									
11									
12									

公式:

=SUMIF(\$A\$2:\$G\$2,H\$2,A3:G3)

微信号: excelpx.tete

2. 单条件求和
3. 单条件模糊求和
4. 多条件模糊求和
5. 多表相同位置求和

公式: B2=SUM(Sheet1:Sheet19!B2)

说明: 在表中间删除或添加表后, 公式结果会自动更新。

Sheet3!B2				
	A	B	C	D
1	产品	本月累计		
2	A	19		
3	B	0		
4	C	0		
5	D	0		
6				

汇总前19个工作表

Sheet17 Sheet18 Sheet19 汇总

6. 按日期和产品求和

(七) 查找与引用公式

1. 单条件查找公式

公式 1: C11=VLOOKUP(B11,B3:F7,4,FALSE)

说明: 查找是 VLOOKUP 最擅长的, 基本用法

	A	B	C	D	E	F
1						
3						
4		姓名	工号	性别	籍贯	出生年月
5		张三丰	KT001	男	北京	1970年8月
6		李四光	KT002	女	天津	1980年9月
7		王麻子	KT003	男	河北	1975年3月
8		赵六儿	KT004	女	河南	1985年12月
9						
10		姓名	籍贯			
11		张三丰	北京			

根据姓名查找籍贯

2. 双向查找公式

公式=INDEX(C3:H7,MATCH(B10,B3:B7,0),MATCH(C10,C2:H2,0))

说明：利用 MATCH 函数查找位置，用 INDEX 函数取值

	A	B	C	D	E	F	G	H
1								
2		姓名	1月	2月	3月	4月	5月	6月
3		刘名	10	54	2	54	5	57
4		吴号码	2	21	21	5	3	48
5		张晴	21	544	20	545	5	545
6		李栋	45	2	35	2	2	1
7		吴风	12	45	21	235	47	54
8								
9		姓名	月份	销售量				
10		张晴	3月	20				
11		李栋	4月					
12								

用公式返回该值

3. 查找最后一条符合条件的记录。

公式：详见下图

说明：0/（条件）可以把不符合条件的变成错误值，而 lookup 可以忽略错误值

	A	B	C	D	E
1					
2	例1: 查找A的最新单价				
3					
4			入库时间	产品名称	入库单价
5			2012-1-2	A	10
6			2012-1-3	B	34
7			2012-1-4	A	19
8			2012-1-5	E	25
9			2012-1-6	A	12
10			2012-1-7	C	25
11					
12					
13		A		12	
14					
15					

C13=LOOKUP(1,0/(C5:C10=B13),D5:D10)

4. 多条件查找

公式：详见下图

说明：公式原理同上一个公式

	A	B	C	D	E	F	G
22	例2: 多条件查找						
23							
24			入库时间	产品名称	入库单价		
25			2012-1-2	A	10		
26			2012-1-3	B	34		
27			2012-1-4	A	19		
28			2012-1-5	E	25		
29			2012-1-6	A	12		
30			2012-1-7	C	25		
31							
32							
33			入库时间	2012-1-4			
34			产品名称	A			
35			入库单价	19			
36							
37							
38							

C35=LOOKUP(1,0/((B25:B30=C33)*(C25:C30=C34)),D25:D30)

5. 指定区域最后一个非空值查找

6. 按数字区域间取对应的值